Usar el ScriptControl para crear aplicaciones extensibles
   Una aplicación "extensible" es aquella a la cual los usuarios pueden agregarle funcionalidad adicional a la prevista por sus diseñadores y constructores. Los productos que integran la suite Office de Microsoft, por ejemplo, son extensibles mediante el uso de Visual Basic para Aplicaciones. Los plugins, los scripts y el uso de XML hacen "extensibles" a los navegadores web.
    Nuestra misión es producir aplicaciones empresariales, que además de cumplir con los requisitos básicos de robustez y "completitud", puedan mercadearse a un precio razonable y que esa "economía de adquisición" sea seguida del más bajo "costo de explotación" posible. Para poder cumplir este objetivo, es necesario alcanzar un gran volumen de ventas. Y para lograr un gran volumen de ventas, en un mercado tan competido como lo es el de los sistemas empresariales, hay que ofrecer algo más que precios bajos. Por esto enfocamos nuestro trabajo de investigación y desarrollo hacia la búsqueda y construcción de sistemas que puedan adaptarse a los requerimientos más específicos de cualquier usuario potencial. Y el resultado fue una solución sencillísima basada en el uso de la tecnología de scripting, y de las posibilidades de interoperabilidad que ofrecen las tecnologías de orientación a objetos.
    El concepto fundamental de nuestro modelo es el de "Punto de Extensión". Un punto de extensión es cualquier operación implementada en el sistema en la cual un usuario podría introducir un cambio. Por supuesto que en casi ninguno de nuestros sistemas "grandes" están identificados los puntos de extensión (al final, cada acción realizada por el sistema es potencialmente un punto de extensión, desde la validación de una caja de texto hasta la agregación de un detalle a una factura). Pero el concepto de punto de función nos permite agregar rápidamente la funcionalidad requerida (o mejor dicho, suministrar una interfaz para que el usuario del sistema agregue la funcionalidad).
    Veamos un primer ejemplo, tomado de uno de nuestros sistemas. Se trata de una transferencia interna de mercancías entre almacenes. Esta operación se inicia desde una forma en la cual el operador introduce los datos básicos de la transferencia: la fecha en la que se realizó, el motivo de la misma, el almacén de origen, el almacén de destino y los detalles (que en esencia se reducen al código y cantidad de los diferentes artículos movidos). No siendo una operación con efecto contable, que se incorporó al sistema sólo como soporte operativo, ignoramos inicialmente el costo unitario de los productos. Pero ocurre que un cliente necesitó conocer este costo, imprimirlo en el documento de transferencia y consultarlo en algunos reportes.
    Para satisfacer este requerimiento con los enfoques tradicionales, habría resultado necesario modificar el código fuente del programa, agregando una referencia a una columna adicional de la tabla de detalles de transferencia (el Costo Unitario). Y este cambio en la fuente habría implicado la necesidad de actualizar la estructura de la tabla de detalles de transferencia en todos los otros usuarios que actualizaran el programa despues de este cambio. Además de que habría sido necesario compilar nuevamente el sistema completo, y tal vez reinstalarlo en los equipos del cliente en caso de que alguno de los componentes empleados en el entorno de desarrollo se hubiese actualizado desde la instalación original.
    Pensando en cómo satisfacer los requerimientos de este cliente (que de todas maneras tuvo que actualizar su sistema), se nos ocurrió insertar en el controlador de la operación, el siguiente código:
	If isOk Then ' isOk indica que el registro se realizó correctamente
  If instTransferencia.Load(instTransferencia.Numero) Then
    ExecuteScript "postTransferencia.vbs", instTransferencia
  End If
End If


    En esencia, lo que hace este código es llamar a la función ExecuteScript, para que esta a su vez determine si existe el archivo postTransferencia.vbs y, de ser así, cargue su código en un control de scripting de Microsoft (Microsoft Script Control), le pase la referencia a instTransferencia (la transferencia que acaba de ser procesada) y ejecute la Sub Main definida en dicho archivo.
    Veamos ahora el codigo contenido en portTransferencia.vbs:
	Sub Main()
Dim dbHandle ' DAO.Database
Dim pItemInv ' Articulo de Inventario
Dim pDetalle ' Detalle de la Transferencia
Dim sQuery ' Para Armar la Sentencia SQL 

  Set dbHandle = Factoria.CrearObjeto("DATABASE")
  Set pItemInv = Factoria.CrearObjeto("ItemInventario")

  For Each pDetalle In OwnerObject.Detalles
    If pItemInv.Load(pDetalle.CodigoProducto) Then
      sQuery = "UPDATE RenglonesTransferencia SET CostoUnitario = " & _    
                   Factoria.NumeroDB(pItemInv.CostoUnitario)
      sQuery = sQuery & " WHERE NumeroTransferencia = " & _
                   Factoria.NumeroDB(OwnerObject.Numero) & _
                   " AND Renglon = " & Factoria.NumeroDB(pDetalle.Renglon)
      On Error Resume Next
      dbHandle.Execute sQuery, 128
      If Err.Number <> 0 Then
        msgBox "ERROR (postTransferencia.vbs):" & vbCrlf & _
                  "Numero: " & Err.Number & vbCrlf & _
                  "Descripción: " & Err.Description
        Exit For
      End If
    End If
  Next
  Set pItemInv = Nothing
  Set dbHandle = Nothing
End Sub


   Por supuesto que este código no tiene nada de especial, salvo por el hecho de que pudo haber sido escrito por cualquier usuario bien informado del sistema sin que nosotros nos hubiésemos siquiera enterado. En esencia, lo que hace es, para cada renglón de la transferencia, agrega a la fila correspondiente de la tabla (identificada por el número de la transferencia y el número del renglón) el costo unitario del artículo (obtenido de la fila correspondiente de la tabla itemsInventario).
    Aunque bien visto, el código si tiene algunas cosas especiales: las referencias a los objetos OwnerObject y Factoria.
    Hasta ahora hemos visto cómo implementar el "punto de extensión" y una extensión concreta (la adición del costoUnitario del artículo correspondiente al renglón de la transferencia). Ahora veremos, en detalle, el simplísimo mecanismo de extensibilidad: la función ExecuteScript:
	Public Sub ExecuteScript(sFileName As String, Optional Caller As Object = Nothing, _
  Optional bNotifyErrors As Boolean = True)
Dim s As String, pF As CFactoria

  If LoadScriptFile(sFileName, pScript, bNotifyErrors) Then
    pScript.AddObject "Factoria", New CFactoria
    If Not Caller Is Nothing Then
      pScript.AddObject "OwnerObject", Caller
    End If
    pScript.Run "Main"
    pScript.Reset
  End If
End Sub 

Public Function LoadScriptFile(sFileName As String, Optional pDest As Object = Nothing, _   
  Optional bNotifyErrors As Boolean = True) As Boolean
Dim s As String, isOk As Boolean, lpScript As Object

  Set lpScript = pDest
  If lpScript Is Nothing Then
    If pScript Is Nothing Then
      On Error Resume Next
      Set pScript = CreateObject("ScriptControl")
      If Err.Number Then ' No está disponible el control de scripting
        Err.Clear
        LoadScriptFile = False
        Exit Function
      End If
    End If ' pScript is Nothing
    Set lpScript = pScript
  End If

  On Error GoTo ErrHandler

  ' Busca el archivo en varias ubicaciones:
  ' Primero, en la carpeta activa
  s = sFileName
  ' si no lo encuentra, lo busca en la carpeta de datos de la ap.
  If Dir(Trim(s)) = "" Then s = BuildDataPath(sFileName, AppName)
  ' luego en la carpeta de la aplicación
  If Dir(Trim(s)) = "" Then s = App.Path & "\" & sFileName
  ' por ultimo, en una ubicación especificada en el registro
  If Dir(Trim(s)) = "" Then
  s = GetSetting(AppName, "General", "UbicacionScripts", "")
  If Right(s, 1) <> "\" Then s = s & "\"
    s = s & sFileName
  End If
  ' Si no existe, sale sin hacer ruido
  If Dir(s) = "" Then
    LoadScriptFile = False
    Exit Function
  End If

  lpScript.Language = "VBScript"
  lpScript.Reset
  s = File2String(s)
  If (s <> "") Then
    isOk = True
    lpScript.AddObject "vsLib", vslib.VSLRTLIB, True
    lpScript.AddCode s
    lpScript.Timeout = -1
  Else
    isOk = False
  End If

ResumePoint:
  LoadScriptFile = isOk
  Exit Function

ErrHandler:
  If bNotifyErrors Then
    ' Errores de sintaxis en el archivo de script
    If lpScript.Error.Number Then
      mAdvertencia "ERROR EN ARCHIVO DE SCRIPT" & vbCrLf & _
          "Archivo: " & sFileName & vbCrLf & _
          "Error: " & lpScript.Error.Description & vbCrLf & _
          "Linea: " & lpScript.Error.Line & vbCrLf & _
          "Texto: " & lpScript.Error.Text
    Else
    ' error de nuestro programa o del entorno
      ReportarError False, Err.Number, Err.Description, "mScriptExecute::LoadScriptFile"
    End If
  End If
  Err.Clear
  isOk = False
  Resume ResumePoint
End Function


   Como es fácil darse cuenta, la mayoría del trabajo de ExecuteScript lo hace LoadScriptFile. Si algo es más aburrido que leer código ajeno, es leer explicaciones sobre el código ajeno, así que comprender los detalles del código (lleno, por demás, de llamadas a funciones de nuestras librerías) queda como ejercicio al lector interesado.
   Lo que si podría resultar útil, es resaltar los puntos clave de todo el fárrago precedente: 

· La utilización de una variable global, llamada pScript. Esta variable se declara As Object para permitir enlace en tiempo de ejecución (late binding) que recomienda el manual del ScriptControl. Todas las llamadas a ExecuteScript utilizan la misma variable global, que es instanciada sólamente una vez. Esto es necesario por que la frecuente creación y destrucción de objetos ScriptControl produce graves "derrames" de memoria. Y es posible porque la ejecución de los scripts es asincrónica (es decir, el proceso anfitrión se detiene mientras el script se ejecuta). 

· La instrucción AddObject nos permite pasarle al ScriptControl (es decir a nuestro código contenido en él) referencias a objetos de la aplicación. La transferencia contenida en instTransferencia en el punto de extensión, es pasada como Caller a ExecuteScript, y este la pasa como OwnerObject a nuestro script. Gracias a los mecanismos de comunicación entre las interfaces de COM (todo Clase en VB publica una interfaz COM), nuestro script puede acceder a las propiedades públicas de los objetos de nuestra aplicación de la misma manera (aunque con una sobrecarga potencialmente significativa) en que lo hacen otros componentes intrínsecos del programa. 

· VSLRTLIB es "la clase" de VSLIB.DLL, una colección de funciones de utilidad general que incorporamos en todos nuestros programas. En ella residen cantidad de funciones como ReportarError, mAdvertencia, BuildDataPath y al menos un centenar de funciones que constituyen nuestro arsenal básico de soluciones prefabricadas. Esta clase está definida como "Global Multiuse", lo que nos permite utilizar sus miembros sin calificación (es decir. escribimos File2String en vez de VSLIB.File2String). El parámetro adicional con valor True en la instrucción donde pasamos VSLRTLIB al script, indica que los miembros públicos de esa clase estarán disponibles sin calificación tambien en el script. 

· Uno de los aspectos más importantes de nuestro método de extensibilidad es la clase CFactoria, de la cual se pasa una instancia con el nombre de Factoria al script, y que -como puede concluirse con sólo ver la frecuencia con la que se utiliza en el script- tiene una importancia primordial en toda esta historia. 

    La Factoria es una fábrica de objetos del espacio de aplicación, que pueden ser utilizados desde el script. Adicionalmente, brinda funciones de soporte general al script (por ejemplo, vemos Factoria.NumeroDB(OwnerObject.Numero). NumeroDB es una función que traduce un argumento numérico a un formato aceptable por el gestro de BD. Idealmente debería estar en una DLL, pero está en un módulo .bas de la aplicación.
    Pero lo realmente importante de la Factoria es la posibilidad de proveer al script de referencias a muchos objetos de la aplicación. La clase CFactoria exporta la función CrearObjeto(sTipoObjeto) As Object (sTipoObjeto debe ser una cadena, pero como VBScript no soporta ningún tipo de datos con excepción del Variant, tenemos por norma no especificar tipos en ninguna función construída para ser llamada desde un script). La estructura de esta función es un Select Case donde, dependiendo del valor de sTipoObjeto se instancian nuevos objetos (o se copian referencias a los objetos existentes, como la referencia a la BD o a los "almacenes": proveedores de servicio de acceso especializados en un tipo específico de objeto) y se pasan al cliente. A través de la factoría los scripts pueden crear y manipular la mayoría de los objetos que definen los niveles de dominio y aplicación del sistema anfitrión. Si quiere ver un ejemplo real de CFactoría, pulse aquí.
    Hay más sobre la extensibilidad: el uso de atributos para agregar datos a la definición básica de las entidades originales del sistema, el uso del registro o de la BD para pasar valores entre dos scripts que se ejecutan secuencialmente, o de DLLs de extensión de las formas. Sobre ellas trataremos en los siguientes capítulos.
    En resumen tenemos que para permitir la extensibilidad lo que hay que hacer es: 

1. Identificar el "punto de extensión" 

2. Colocar en el punto de extensión la activación del mecanismo de extensibilidad. Un posible mecanismo de extensibilidad es el descrito por las funciones LoadScriptFile y ExecuteScript. Seguramente bastará con pensar un poco para encontrar mecanismos más eficaces o más adaptados al estilo o necesidades de cada desarrollador. 

3. Incluir en el mecanismo de extensibilidad la posibilidad de que el cliente acceda a los objetos del dominio y de la aplicación. 

 

